

INDEX & MATCH Function LAB

Worksheet: Sales Data

Product ID	Product	Category	Jan Sales	Feb Sales	Mar Sales	Apr Sales	May Sales
101	PRODA	Electronics	120	130	140	150	160
102	PRODB	Furniture	150	160	170	180	190
103	PRODC	Electronics	200	210	220	230	240
104	PRODD	Clothing	90	100	110	120	130
105	PRODE	Furniture	220	230	240	250	260
106	PRODF	Electronics	130	140	150	160	170

Ques1: Use INDEX and MATCH to find the sales for Product C in March.

Solution:

Step1: Identify the cell range containing the product names and sales data.

Step2: Use MATCH to find the row for Product C. Use "PRODC" as the lookup value, B2:B7 as the lookup array, and 0 for an exact match: **MATCH(J5,B2:B7,0)**.

Step 3: Use MATCH again to find the column for March Sales. Use "Mar Sales" as the lookup value, C1:G1 as the lookup array, and 0 for an exact match: **MATCH(K4,D1:H1,0)**.

Step 4: Use INDEX to get the value at the intersection of the row and column found above:
=INDEX(D2:H7,MATCH(J5,B2:B7,0),MATCH(K4,D1:H1,0)).

Product	Mar Sales
PRODC	220

Ques2: Use INDEX and MATCH to find the category for Product E.

Solution:

Step 1: Use MATCH to find the row number for Product E. Use "PRODE" as the lookup value and B2:B7 as the lookup array: **MATCH(J9,B2:B7,0)**.

Step 2: Use INDEX to get the category from the corresponding row in the C column
=INDEX(C2:C7,MATCH(J9,B2:B7,0)).

Product	Category
PRODE	Furniture

Ques3: Use INDEX and MATCH to find the maximum sales for Product B across all months.

Solution:

Step 1: Use `MATCH` to find the row number for Product B. Use "PRODB" as the lookup value and B2:B7 as the lookup array: `MATCH(J13,B2:B7,0)`.

Step 2: Use `INDEX` to get the row of sales data for Product B and `MAX` to find the maximum value across the row:

`=MAX(INDEX(D2:H7,MATCH(J13,B2:B7,0),0))`.

Product	Max Sales
PRODB	190

Ques4: Use INDEX and MATCH to find the month with the maximum sales for Product A.

Solution:

Step 1: Use `MATCH` to find the row number for Product A. Use "PRODA" as the lookup value and B2:B7 as the lookup array: `MATCH(J16,B2:B7,0)`.

Step 2: Use `INDEX` to get the row of sales data for Product A by using the formula:

`(INDEX(D2:H7,MATCH(J16,B2:B7,0),0))`.

Step 3: Use `MAX` to find the highest sales figure for Product A in that row by using the formula:

`(MAX(INDEX(D2:H7,MATCH(J16,B2:B7,0),0)),INDEX(D2:H7,MATCH(J16,B2:B7,0),0),0))`.

Step 4: Use `MATCH` to locate the column of this maximum value by using the formula:

`MATCH(MAX(INDEX(D2:H7,MATCH(J16,B2:B7,0),0)),INDEX(D2:H7,MATCH(J16,B2:B7,0),0),0))`.

Step 5: Use `INDEX` to return the month from the headers by using the formula:

`=INDEX(D1:H1,MATCH(MAX(INDEX(D2:H7,MATCH(J16,B2:B7,0),0)),INDEX(D2:H7,MATCH(J16,B2:B7,0),0),0))`.

Product	Max Sales
PRODA	May Sales

Ques5: Use INDEX, MATCH, and SUMIF to sum the sales for all products in the "Electronics" category for April.

Solution:

Step 1: Use `MATCH` to dynamically find the April column: `MATCH(B21,D1:H1,0)`.

Step 2: Use `INDEX` function to get the values for April Month by using the formula:

`INDEX(D1:H7,0,MATCH(B21,D1:H1,0))`.

Step 3: Use `SUMIF` to find the sum of sales for Electronics Category by using the formula:

=SUMIF(C1:C7,A22,INDEX(D1:H7,0,MATCH(B21,D1:H1,0))).

Category	Apr Sales
Electronics	540

Ques6: Use INDEX and MATCH to calculate the average sales for Product D across all months.

Solution:

Step 1: Use MATCH to find the row for Product D. Use "PRODD" as the lookup value and B2 : B7 as the lookup array: **MATCH(J22,B2:B7,0).**

Step 2: Use INDEX to get all sales data for Product D, then use AVERAGE to calculate the average:

=AVERAGE(INDEX(D2:H7,MATCH(J22,B2:B7,0),0)).

Product	Average Sales
PRODD	110

Ques7: Use INDEX and MATCH to find the sales for Product ID 105 in May.

Solution:

Step 1: Use MATCH to locate the row for Product ID 105: **MATCH(J25,A2:A7,0).**

Step 2: Use MATCH to locate the May Sales column: **MATCH(K24,D1:H1,0).**

Step 3: Use INDEX to retrieve the value at the intersection:

=INDEX(D2:H7,MATCH(J25,A2:A7,0),MATCH(K24,D1:H1,0)).

Product ID	May Sales
105	260

Ques8: Use INDEX and MATCH to create a dynamic lookup where the user can input a product and a month, and the formula returns the corresponding sales.

Solution:

Step 1: Create a table for the user to get the Product name(Q12) and Month(R12) to get the sales.

Step 2: Use MATCH to find the row for the product name entered in (Q12) : **MATCH (Q13 ,B1 : B7 , 0) .**

Step 3: Use `MATCH` to find the column for the month entered in (R12) by combining it with "Sales": `MATCH(R13,D1:H1,0)`.

Step 4: Use `INDEX` to retrieve the value at the intersection:

`=INDEX(D1:H7,MATCH(Q13,B1:B7,0),MATCH(R13,D1:H1,0))`.

Product	Month	Sales
PRODC	Feb Sales	210